```
<< KnotTheory`
```

```
Loading KnotTheory` version of February 5, 2013, 3:48:46.4762.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
(*WP:Wedge Product*)WSort[expr_] := Expand[expr /. w_W :> Signature[w] * Sort[w]];
WP[0, _] = WP[_, 0] = 0;
WP[a_, b_] :=
  WSort[Distribute[a ** b] /. (c1_. * w1_W) ** (c2_. * w2_W) :> c1 c2 Join[w1, w2]];

(*IM:Interior Multiplication*)
IM[{}, expr_] := expr;
IM[i_, w_W] :=
  If[FreeQ[w, i], 0, -(-1)^Position[w, i][[1, 1]] * DeleteCases[w, i]];
IM[{is___, i_}, w_W] := IM[{is}, IM[i, w]];
IM[is_List, expr_] := expr /. w_W :> IM[is, w]

(*pA on Crossings*)
pA[Xp[i_, j_, k_, l_]] := AHD[(t[i] == t[k]) (t[j] == t[l]), {i, l},
   W[j, k], W[l, i] + (t[i] - 1) W[l, j] - t[l] W[l, k] + W[i, j] + t[l] W[j, k]];
pA[Xm[i_, j_, k_, l_]] := AHD[(t[i] == t[k]) (t[j] == t[l]), {i, j}, W[k, l],
  t[j] W[i, j] - t[j] W[i, l] + W[j, k] + (t[i] - 1) W[j, l] + W[k, l]]

(*Variable Equivalences*)
ReductionRules[Times[]] = {};
ReductionRules[Equal[a_, b__]] := (# → a) & /@ {b};
ReductionRules[eqs_Times] := Join @@ (ReductionRules /@ List @@ eqs)

(*AHD:Alexander Half Densities*)
AHD[eqs_, is_, -os_, p_] := AHD[eqs, is, os, Expand[-p]];
AHD /: Reduce[AHD[eqs_, is_, os_, p_]] :=
  AHD[eqs, Sort[is], WSort[os], WSort[p /. ReductionRules[eqs]]];
AHD /: AHD[eqs1_, is1_, os1_, p1_] AHD[eqs2_, is2_, os2_, p2_] :=
 Module[{glued = Intersection[Union[is1, is2], List @@ Union[os1, os2]]},
  Reduce[AHD[eqs1 * eqs2 //. eq1_Equal * eq2_Equal /;
       Intersection[List @@ eq1, List @@ eq2] =!= {} :> Union[eq1, eq2], Complement[
     Union[is1, is2], glued], IM[glued, WP[os1, os2]], IM[glued, WP[p1, p2]]]]]

(*pA on Circuit Diagrams*)
pA[cd_CircuitDiagram, eqs___] := pA[cd, {}, AHD[Times[eqs], {}, W[], W[]]];
pA[cd_CircuitDiagram, done_, ahd_AHD] :=
  Module[{pos = First[Ordering[Length[Complement[List @@ #, done]] & /@ cd]]},
   pA[Delete[cd, pos], Union[done, List @@ cd[[pos]]], ahd * pA[cd[[pos]]]]];
pA[CircuitDiagram[], _, ahd_AHD] := ahd
```

```
pd = Delete[PD[Knot[10, 100]], 5]
```

KnotTheory::loading : Loading precomputed data in PD4Knots`.

```
PD[X[6, 2, 7, 1], X[18, 6, 19, 5], X[20, 13, 1, 14], X[14, 7, 15, 8], X[16, 9, 17, 10],
 X[4, 11, 5, 12], X[8, 15, 9, 16], X[12, 19, 13, 20], X[2, 18, 3, 17]]
```

```
cd = CircuitDiagram @@ pd /. {x_X?PositiveQ ⧴ Xp @@ x, x_X?NegativeQ ⧴ Xm @@ x}
```

```
CircuitDiagram[Xp[6, 2, 7, 1], Xp[18, 6, 19, 5],
 Xm[20, 13, 1, 14], Xm[14, 7, 15, 8], Xm[16, 9, 17, 10],
 Xm[4, 11, 5, 12], Xm[8, 15, 9, 16], Xm[12, 19, 13, 20], Xp[2, 18, 3, 17]]
```

```
cd // pA
```

```
AHD[(t[4] == t[5] == t[6] == t[7] == t[8] == t[9] == t[10])
   (t[1] == t[2] == t[3] == t[11] == t[12] == t[13] == t[14] == t[15] ==
     t[16] == t[17] == t[18] == t[19] == t[20]), {4, 11}, W[3, 10],
 -t[1]^2 W[3, 4] + 2 t[1]^3 W[3, 4] - 2 t[1]^4 W[3, 4] + t[1]^5 W[3, 4] + t[1]^2 t[4] W[3, 4] -
  4 t[1]^3 t[4] W[3, 4] + 4 t[1]^4 t[4] W[3, 4] - 3 t[1]^5 t[4] W[3, 4] + t[1]^6 t[4] W[3, 4] +
  2 t[1]^3 t[4]^2 W[3, 4] - 3 t[1]^4 t[4]^2 W[3, 4] + 2 t[1]^5 t[4]^2 W[3, 4] -
  t[1]^6 t[4]^2 W[3, 4] + t[1] W[3, 10] - 2 t[1]^2 W[3, 10] + 2 t[1]^3 W[3, 10] -
  t[1]^4 W[3, 10] - t[1] t[4] W[3, 10] + 3 t[1]^2 t[4] W[3, 10] - 2 t[1]^3 t[4] W[3, 10] +
  2 t[1]^4 t[4] W[3, 10] - t[1]^5 t[4] W[3, 10] - t[1]^2 t[4]^2 W[3, 10] + t[1]^3 t[4]^2 W[3, 10] -
  t[1]^4 t[4]^2 W[3, 10] + t[1]^5 t[4]^2 W[3, 10] - t[1] W[3, 11] + 2 t[1]^2 W[3, 11] -
  2 t[1]^3 W[3, 11] + t[1]^4 W[3, 11] + 2 t[1] t[4] W[3, 11] - 5 t[1]^2 t[4] W[3, 11] +
  5 t[1]^3 t[4] W[3, 11] - 4 t[1]^4 t[4] W[3, 11] + t[1]^5 t[4] W[3, 11] - t[1] t[4]^2 W[3, 11] +
  4 t[1]^2 t[4]^2 W[3, 11] - 5 t[1]^3 t[4]^2 W[3, 11] + 5 t[1]^4 t[4]^2 W[3, 11] -
  2 t[1]^5 t[4]^2 W[3, 11] - t[1]^2 t[4]^3 W[3, 11] + 2 t[1]^3 t[4]^3 W[3, 11] -
  2 t[1]^4 t[4]^3 W[3, 11] + t[1]^5 t[4]^3 W[3, 11] - t[1]^2 W[4, 10] + 3 t[1]^3 W[4, 10] -
  4 t[1]^4 W[4, 10] + 3 t[1]^5 W[4, 10] - t[1]^6 W[4, 10] + t[1]^2 t[4] W[4, 10] -
  4 t[1]^3 t[4] W[4, 10] + 6 t[1]^4 t[4] W[4, 10] - 6 t[1]^5 t[4] W[4, 10] +
  4 t[1]^6 t[4] W[4, 10] - t[1]^7 t[4] W[4, 10] + t[1]^3 t[4]^2 W[4, 10] - 3 t[1]^4 t[4]^2 W[4, 10] +
  4 t[1]^5 t[4]^2 W[4, 10] - 3 t[1]^6 t[4]^2 W[4, 10] + t[1]^7 t[4]^2 W[4, 10] -
  t[1]^2 t[4] W[4, 11] + t[1]^3 t[4] W[4, 11] - t[1]^4 t[4] W[4, 11] + t[1]^5 t[4] W[4, 11] +
  t[1]^2 t[4]^2 W[4, 11] - 2 t[1]^3 t[4]^2 W[4, 11] + 2 t[1]^4 t[4]^2 W[4, 11] -
  3 t[1]^5 t[4]^2 W[4, 11] + t[1]^6 t[4]^2 W[4, 11] + t[1]^3 t[4]^3 W[4, 11] -
  2 t[1]^4 t[4]^3 W[4, 11] + 2 t[1]^5 t[4]^3 W[4, 11] - t[1]^6 t[4]^3 W[4, 11] + t[1] W[10, 11] -
  3 t[1]^2 W[10, 11] + 4 t[1]^3 W[10, 11] - 3 t[1]^4 W[10, 11] + t[1]^5 W[10, 11] -
  t[1] t[4] W[10, 11] + 5 t[1]^2 t[4] W[10, 11] - 7 t[1]^3 t[4] W[10, 11] +
  7 t[1]^4 t[4] W[10, 11] - 5 t[1]^5 t[4] W[10, 11] + t[1]^6 t[4] W[10, 11] -
  2 t[1]^2 t[4]^2 W[10, 11] + 5 t[1]^3 t[4]^2 W[10, 11] - 6 t[1]^4 t[4]^2 W[10, 11] +
  6 t[1]^5 t[4]^2 W[10, 11] - 2 t[1]^6 t[4]^2 W[10, 11] - t[1]^3 t[4]^3 W[10, 11] +
  2 t[1]^4 t[4]^3 W[10, 11] - 2 t[1]^5 t[4]^3 W[10, 11] + t[1]^6 t[4]^3 W[10, 11]]
```

```
IEM[W[], expr_] := expr;  (*Interior-Exterior Multiplication*)
IEM[W[i_, is___], expr_] := IEM[W[is], IM[{i}, expr] + WP[W[i], expr]];
```

```
IEM[W[3, 11], W[3, 10]]
```

```
-W[10, 11]
```

```
IEM[W[3, 10], W[3, 10]]
```

```
W[]
```

```
IEM[W[1, 2, 3], W[1, 2, 3]]
```

```
W[]
```

```
IEM[W[1, 2, 3, 4], W[1, 2, 3, 4]]
```

```
W[]
```

```
Signature[Reverse[Range[#]]] & /@ Range[10]
```

```
{1, -1, -1, 1, 1, -1, -1, 1, 1, -1}
```

```
HS[AHD[eqs_, is_, os_, p_]] := AEDF[eqs, is, os, p /. w_W :> IEM[w, os]]
```

```
cd // pA // HS
```

```
AEDF[(t[4] == t[5] == t[6] == t[7] == t[8] == t[9] == t[10])
   (t[1] == t[2] == t[3] == t[11] == t[12] == t[13] == t[14] == t[15] ==
     t[16] == t[17] == t[18] == t[19] == t[20]), {4, 11}, W[3, 10],
 t[1] W[] - 2 t[1]^2 W[] + 2 t[1]^3 W[] - t[1]^4 W[] - t[1] t[4] W[] + 3 t[1]^2 t[4] W[] -
  2 t[1]^3 t[4] W[] + 2 t[1]^4 t[4] W[] - t[1]^5 t[4] W[] - t[1]^2 t[4]^2 W[] + t[1]^3 t[4]^2 W[] -
  t[1]^4 t[4]^2 W[] + t[1]^5 t[4]^2 W[] + t[1]^2 W[3, 4] - 3 t[1]^3 W[3, 4] + 4 t[1]^4 W[3, 4] -
  3 t[1]^5 W[3, 4] + t[1]^6 W[3, 4] - t[1]^2 t[4] W[3, 4] + 4 t[1]^3 t[4] W[3, 4] -
  6 t[1]^4 t[4] W[3, 4] + 6 t[1]^5 t[4] W[3, 4] - 4 t[1]^6 t[4] W[3, 4] + t[1]^7 t[4] W[3, 4] -
  t[1]^3 t[4]^2 W[3, 4] + 3 t[1]^4 t[4]^2 W[3, 4] - 4 t[1]^5 t[4]^2 W[3, 4] + 3 t[1]^6 t[4]^2 W[3, 4] -
  t[1]^7 t[4]^2 W[3, 4] + t[1] W[3, 11] - 3 t[1]^2 W[3, 11] + 4 t[1]^3 W[3, 11] -
  3 t[1]^4 W[3, 11] + t[1]^5 W[3, 11] - t[1] t[4] W[3, 11] + 5 t[1]^2 t[4] W[3, 11] -
  7 t[1]^3 t[4] W[3, 11] + 7 t[1]^4 t[4] W[3, 11] - 5 t[1]^5 t[4] W[3, 11] + t[1]^6 t[4] W[3, 11] -
  2 t[1]^2 t[4]^2 W[3, 11] + 5 t[1]^3 t[4]^2 W[3, 11] - 6 t[1]^4 t[4]^2 W[3, 11] +
  6 t[1]^5 t[4]^2 W[3, 11] - 2 t[1]^6 t[4]^2 W[3, 11] - t[1]^3 t[4]^3 W[3, 11] +
  2 t[1]^4 t[4]^3 W[3, 11] - 2 t[1]^5 t[4]^3 W[3, 11] + t[1]^6 t[4]^3 W[3, 11] - t[1]^2 W[4, 10] +
  2 t[1]^3 W[4, 10] - 2 t[1]^4 W[4, 10] + t[1]^5 W[4, 10] + t[1]^2 t[4] W[4, 10] -
  4 t[1]^3 t[4] W[4, 10] + 4 t[1]^4 t[4] W[4, 10] - 3 t[1]^5 t[4] W[4, 10] + t[1]^6 t[4] W[4, 10] +
  2 t[1]^3 t[4]^2 W[4, 10] - 3 t[1]^4 t[4]^2 W[4, 10] + 2 t[1]^5 t[4]^2 W[4, 10] -
  t[1]^6 t[4]^2 W[4, 10] + t[1] W[10, 11] - 2 t[1]^2 W[10, 11] + 2 t[1]^3 W[10, 11] -
  t[1]^4 W[10, 11] - 2 t[1] t[4] W[10, 11] + 5 t[1]^2 t[4] W[10, 11] - 5 t[1]^3 t[4] W[10, 11] +
  4 t[1]^4 t[4] W[10, 11] - t[1]^5 t[4] W[10, 11] + t[1] t[4]^2 W[10, 11] -
  4 t[1]^2 t[4]^2 W[10, 11] + 5 t[1]^3 t[4]^2 W[10, 11] - 5 t[1]^4 t[4]^2 W[10, 11] +
  2 t[1]^5 t[4]^2 W[10, 11] + t[1]^2 t[4]^3 W[10, 11] - 2 t[1]^3 t[4]^3 W[10, 11] +
  2 t[1]^4 t[4]^3 W[10, 11] - t[1]^5 t[4]^3 W[10, 11] - t[1]^2 t[4] W[3, 4, 10, 11] +
  t[1]^3 t[4] W[3, 4, 10, 11] - t[1]^4 t[4] W[3, 4, 10, 11] + t[1]^5 t[4] W[3, 4, 10, 11] +
  t[1]^2 t[4]^2 W[3, 4, 10, 11] - 2 t[1]^3 t[4]^2 W[3, 4, 10, 11] + 2 t[1]^4 t[4]^2 W[3, 4, 10, 11] -
  3 t[1]^5 t[4]^2 W[3, 4, 10, 11] + t[1]^6 t[4]^2 W[3, 4, 10, 11] + t[1]^3 t[4]^3 W[3, 4, 10, 11] -
  2 t[1]^4 t[4]^3 W[3, 4, 10, 11] + 2 t[1]^5 t[4]^3 W[3, 4, 10, 11] - t[1]^6 t[4]^3 W[3, 4, 10, 11]]
```

```
?? PositiveQ
```

PositiveQ[xing] returns True if xing is a positive (right handed) crossing and False if it is negative (left handed).

```
PositiveQ[X[KnotTheory`Private`i_, KnotTheory`Private`j_, KnotTheory`Private`k_,
    KnotTheory`Private`l_]] /; KnotTheory`Private`j - KnotTheory`Private`l == 1 ||
   KnotTheory`Private`l - KnotTheory`Private`j > 1 = True

PositiveQ[X[KnotTheory`Private`i_, KnotTheory`Private`j_, KnotTheory`Private`k_,
    KnotTheory`Private`l_]] /; KnotTheory`Private`l - KnotTheory`Private`j == 1 ||
   KnotTheory`Private`j - KnotTheory`Private`l > 1 = False

PositiveQ[_Xp] = True

PositiveQ[_Xm] = False
```